Paper 199-30

Testing the Hypothesis of a Kroneckar Product Covariance Matrix in Multivariate Repeated Measures Data

Anuradha Roy, The University of Texas at San Antonio, San Antonio, TX
Ravindra Khattree, Oakland University, Rochester, MI

ABSTRACT

We consider the problem of testing of a Kronecker product structured variance covariance matrix against the unstructured variance covariance matrix in the context of multivariate repeated measures data. A likelihood ratio test statistic is developed for this purpose. However, the test statistic for this test cannot be expressed in a closed form. An algorithm, is suggested for the maximum likelihood estimation of the variance covariance matrices to test the proposed hypothesis. The algorithm, developed by using *IML* procedure of SAS®, successfully converged in all the data sets on which it was tried. The proposed hypothesis can also be in principle, tested in two stages by the use of *MIXED* procedure of SAS. However, it was our experience that *PROC MIXED* often fails to converge. Three real data examples illustrating the algorithm and calculations are also presented.

INTRODUCTION

Data that contain multiple measurements over time on the same response variable for each subject or experimental unit are very common, in many fields such as biomedical, pharmaceutical, industrial engineering, business etc. This type of data are commonly called (univariate) repeated measures data. Multivariate repeated measures data or doubly multivariate data are those where multiple measurements are made over time on more than one response variable on each subject or unit. Suppose there are q response variables and on each response variable, observations are taken over p time points. We represent information on a typical subject by $\boldsymbol{y}$, a $pq \times 1$ dimensional column vector obtained by stacking all q responses at the first time point, then stacking all q responses at the second time point below it and so on. Assume that $\boldsymbol{y}$ follows a multivariate normal distribution with mean $\boldsymbol{\mu}$ and with a $pq \times pq$ positive definite variance covariance matrix $\boldsymbol{\Omega}$. Without any assumption on covariance structures, $\boldsymbol{\Omega}$ has $\frac{pq(pq+1)}{2}$ number of unknown parameters. Estimation of $\boldsymbol{\Omega}$ is not possible when the number of subjects, say, n is less than or equal to pq. This is a common scenario, especially when p is large. In order to circumvent this problem one may assume some covariance structure on $\boldsymbol{\Omega}$ and a convenient common structure that can be assumed is the Kronecker product structure. Several authors, e.g, Boik (1991), Galecki (1994), Naik and Rao (2001), and Chaganty and Naik (2002) have used this Kronecker product structure variance covariance matrix in their analyses. Roy (2002), Roy and Khattree (2003, 2005) have recently used this structure in the classification problems.

To be specific, we assume $\boldsymbol{\Omega}$ to be of the form

$$\boldsymbol{\Omega} = \boldsymbol{V} \otimes \boldsymbol{\Sigma}$$

where $\boldsymbol{V}$ is a $p \times p$ symmetric positive definite and $\boldsymbol{\Sigma}$ is $q \times q$ also symmetric positive definite. The matrix $\boldsymbol{V}$ represents the correlation between repeated measures on a given subject and for a given response variable. Likewise, $\boldsymbol{\Sigma}$ represents the covariance between all response variables on a given subject and for a given time point. The above covariance structure makes an implicit assumption that for all the response variables, the correlation structure between repeated measures remains the same and that for covariance between all the response variables does not depend on time and remains constant for all time points.

The matrix of correlation of repeated measures $\boldsymbol{V}$, for a given response variables, can have any structure or can even be unstructured. Two covariance structures for $\boldsymbol{V}$ are very common. These are, the

compound symmetric (CS) structure or the autoregressive covariance structure of order 1 $\big(\text{AR}(1)\big)$. The compound symmetric covariance structure assumes that all repeated measures are equicorrelated and do not depend on the duration between the two time points. However the AR(1) is more realistic and is often assumed in situations when data are collected at equispaced time intervals and where observations close to each other in time duration are likely to be more closely associated. In this paper we will only consider $\boldsymbol{V}$ as AR(1). The other case of compound symmetry follows essentially on similar lines.

Motivated by above discussion, we will consider the SAS implementation of likelihood ratio tests for the following hypothesis:

$$H_o : \boldsymbol{\Omega} = \boldsymbol{V} \otimes \boldsymbol{\Sigma},\ \boldsymbol{V}\ \text{AR}(1) \quad \text{against} \quad H_a : \boldsymbol{\Omega}\ \text{unstructured}.$$

Since $\boldsymbol{V}$ has an AR(1) structure,

$$\boldsymbol{V} = \begin{bmatrix} 1 & \rho & \rho^2 & \cdots & \rho^{p-1} \\ \rho & 1 & \rho & \cdots & \rho^{p-2} \\ \rho^2 & \rho & 1 & \cdots & \rho^{p-3} \\ . & . & . & \cdots & . \\ . & . & . & \cdots & . \\ \rho^{p-1} & \rho^{p-2} & \rho^{p-3} & \cdots & 1 \end{bmatrix}.$$

It is well known that the determinant of this matrix is given by

$$|\boldsymbol{V}| = \big(1-\rho^2\big)^{p-1},$$

and its inverse is

$$\boldsymbol{V}^{-1} = (1-\rho^2)^{-1}[\rho^2 \boldsymbol{C}_1 + \rho \boldsymbol{C}_2 + \boldsymbol{I}_p],$$

where $\boldsymbol{C}_1 = \text{diag}(0, 1, \ldots, 1, 0)$ and $\boldsymbol{C}_2$ is a tridiagonal matrix with 0 on the diagonal and 1 on the first superdiagonal and on the first subdiagonal.

THE LIKELIHOOD RATIO TEST

The likelihood ratio test compares the maximum value of the likelihood function L restricted to the region defined by the null hypothesis H_o, to the maximum of likelihood function L in the unrestricted region, H_a. Thus the ratio

$$\Lambda = \frac{\max_{H_o} L}{\max_{H_a} L}$$

or a function of it, is used as the test statistic to test the null hypothesis H_o. It is well known that, for large samples and under normality assumption $L = -2\ln\Lambda$ is approximately χ^2_ν under H_o where the degrees of freedom (d.f.) ν is equal to the number of parameters estimated under H_a minus the number of parameters estimated under H_o.

It must be pointed out that this hypothesis can also be, in principle (Khattree and Naik, 1999), tested by using *PROC MIXED* of SAS (SAS Institute Inc., 2004), using a *REPEATED* statement to identify the correlated observations and to impose a covariance structure on the responses, collected on the same subject or the unit. To compute the test statistic $-2\ln\Lambda$, given by

$$-2\ln\Lambda = \left[-2\ln\max_{H_o} L(\boldsymbol{\mu}, \boldsymbol{V}, \boldsymbol{\Sigma}; \boldsymbol{Y})\right] - \left[-2\ln\max_{H_a} L(\boldsymbol{\mu}, \boldsymbol{\Omega}; \boldsymbol{Y})\right],$$

the likelihood functions under both structured $H_o\big(\boldsymbol{\Omega} = \boldsymbol{V}\otimes\boldsymbol{\Sigma}, \boldsymbol{V}\ \text{AR}(1)\big)$ and under $H_a(\boldsymbol{\Omega}$ unstructured), are to be maximized separately. [1] This can be done by using the option *METHOD=ML* in *PROC MIXED* statement which will implement the maximum likelihood procedure for the estimates of unknown parameters. The options *TYPE=UN⊗AR(1)* and *TYPE=UN* are used to calculate the "-2 Log

[1] SAS uses the notation $\boldsymbol{\Sigma} \otimes \boldsymbol{V}$ for $\boldsymbol{V} \otimes \boldsymbol{\Sigma}$.

Likelihood" for the covariance structure under H_o, when $\boldsymbol{V}$ is AR(1) and H_a respectively.

The SAS code using *PROC MIXED*, partially obtained from Khattree and Naik (1999) and applied on the Dental dataset A, described later to fit the above models is given in the Appendix 2. In the program the data are written in the univariate form. The variable *TIME* represents the repeated measurements and *MVAR* represents the characteristics of the data set. Both *TIME* and *MVAR* are specified in the *REPEATED* statement to justify that the data is multivariate in two directions. The selected parts of the corresponding output are also given in the Appendix 3.

Although the above approach is found to be algorithmically successful in many applications such as data sets given in Khattree and Naik (1999), it has been our experience that in many more complex applications, especially those dealing with the Kronecker product covariance structure, the algorithm used by *PROC MIXED* often does not converge and stops because the likelihood diverges. As a result the likelihood test statistic cannot be obtained. This necessitates another alternative algorithm for maximum likelihood estimates of $\boldsymbol{V}, \boldsymbol{\Sigma}$ and $\boldsymbol{\Omega}$. Thus we provide a SAS- IML algorithm in Appendix 1 for this purpose. Our algorithm successfully converges for all the data sets that we considered.

Specifically, our algorithm uses the explicit or iterative formulas given in Roy (2002). Brief details are given here. Under H_a, the calculations result in

$$\hat{\boldsymbol{\Omega}}_{H_a} = \frac{1}{n}\boldsymbol{S},$$

where $\boldsymbol{S}$ is the $pq \times pq$ dimensional sample variance covariance matrix and n is the number of random samples taken from the population. However, under H_o,

$$\hat{\boldsymbol{\Omega}}_{H_o} = \hat{\boldsymbol{V}}_{H_o} \otimes \hat{\boldsymbol{\Sigma}}_{H_o},$$

where $\hat{\boldsymbol{V}}_{H_o}$ and $\hat{\boldsymbol{\Sigma}}_{H_o}$ are the maximum likelihood estimates of $\boldsymbol{V}$ and $\boldsymbol{\Sigma}$ respectively. It must be emphasized that $\boldsymbol{V}$, having the AR(1) structure is a function of ρ which must be estimated along with $\boldsymbol{\Sigma}$. The maximum likelihood estimates $\hat{\boldsymbol{\Sigma}}_{H_o}$ and $\hat{\rho}$ are obtained by simultaneously and iteratively solving the following equations (1) and (2). Thus, the maximum likelihood estimate $\hat{\boldsymbol{V}}_{H_o}$ is obtained from $\boldsymbol{V}$ by replacing ρ by $\hat{\rho}$.

$$\hat{\boldsymbol{\Sigma}}_{H_o} = \frac{1}{np}\sum_{i=1}^{n}\sum_{l=1}^{p}\sum_{m=1}^{p} v^{lm}\,(\boldsymbol{y}_{im} - \bar{\boldsymbol{y}}_m)(\boldsymbol{y}_{il} - \bar{\boldsymbol{y}}_l)', \tag{1}$$

and

$$-2n(p-1)q\rho^3 + b\rho^2 + [2n(p-1)q - 2(c+a)]\rho + b = 0, \tag{2}$$

where $a = \mathrm{tr}(\boldsymbol{C}_1 \otimes \boldsymbol{\Sigma}^{-1})\boldsymbol{S}$, $b = \mathrm{tr}(\boldsymbol{C}_2 \otimes \boldsymbol{\Sigma}^{-1})\boldsymbol{S}$ and $c = \mathrm{tr}(\boldsymbol{I}_p \otimes \boldsymbol{\Sigma}^{-1})\boldsymbol{S}$.

Therefore, substituting the corresponding maximum likelihood estimates of the parameters into the likelihood function, the likelihood ratio is given by,

$$\Lambda = \frac{|\hat{\boldsymbol{V}}_{H_o}|^{-\frac{qn}{2}}|\hat{\boldsymbol{\Sigma}}_{H_o}|^{-\frac{pn}{2}}e^{-\frac{1}{2}\mathrm{tr}(\hat{\boldsymbol{V}}_{H_o}\otimes\hat{\boldsymbol{\Sigma}}_{H_o})^{-1}\boldsymbol{S}}}{|\boldsymbol{S}|^{-\frac{n}{2}}n^{\frac{npq}{2}}e^{-\frac{npq}{2}}}.$$

The degrees of freedom ν for the null distribution of $-2\ln\Lambda$ is given by,

$$\nu = \frac{pq(pq+1)}{2} - \frac{q(q+1)}{2} - 1.$$

THE ALGORITHM

From above, it is clear that the sample variance covariance matrix $\boldsymbol{S}$ can be computed in a straight

forward way, but, $\hat{\boldsymbol{V}}_{H_o}$ and $\hat{\boldsymbol{\Sigma}}_{H_o}$ need to be computed iteratively. The following iterative steps are suggested to get the maximum likelihood estimates of $\hat{\boldsymbol{V}}_{H_o}$ and $\hat{\boldsymbol{\Sigma}}_{H_o}$.
Step 1: Get the initial pooled sample variance covariance matrix for repeated measures. Say it is $\boldsymbol{G}^{\star}$. Get the average of the 1st superdiagonal elements of $\boldsymbol{G}^{\star}$, say, $\rho_{1\star}$. Then get the average of the 2nd superdiagonal elements of $\boldsymbol{G}^{\star}$, say, $\rho_{2\star}$ and so on. The initial estimate of ρ is obtained as $\hat{\rho}_o = \left(\frac{(\rho_{1\star})^{p-1} + (\rho_{2\star})^{\frac{p-1}{2}} + (\rho_{3\star})^{\frac{p-1}{3}} + \cdots + \rho_{p-1\,\star}}{p-1} \right)^{\frac{1}{p-1}}$, and thus $\hat{\boldsymbol{V}}_o$, an initial estimate of $\boldsymbol{V}$ is obtained by replacing ρ by $\hat{\rho}_o$.
Step 2: Compute $\hat{\boldsymbol{\Sigma}}_{H_o}$ from the Equation (1).
Step 3: Compute a, b and c using $\hat{\boldsymbol{\Sigma}}_{H_o}$ from *Step 2*.
Step 4: Compute the value of $\hat{\boldsymbol{\rho}}$ by solving the cubic Equation (2).
Step 5: Compute the revised estimate of $\hat{\boldsymbol{V}}_{H_o}$ from $\hat{\boldsymbol{\rho}}$.
Step 6: Compute the revised estimate of $\hat{\boldsymbol{\Sigma}}_{H_o}$ from Equation (1), using $\hat{\boldsymbol{V}}_{H_o}$ obtained in *Step 5*.
Step 7: Repeat Steps 3, 4, 5 and 6 until convergence is attained. This is ensured by verifying that the maximum of the absolute difference between two successive values of $\hat{\boldsymbol{\rho}}$ and the absolute difference between two successive values of $\mathrm{tr}(\hat{\boldsymbol{\Sigma}}_{H_o})$ is less than a pre-determine number $\epsilon\,(= 10^{-6}, \text{say})$.

The SAS- IML code for above algorithm is given in the Appendix 1.

ILLUSTRATIVE EXAMPLES

In this section, we illustrate our algorithm with a few simple real world examples. Further for several special cases (by substituting the number of response variables equal to one) the algorithm was tried on many standard univariate repeated measures datasets, including those used in Khattree and Naik (1999). Since for these datasets and special cases our algorithm (*PROC IML*) always converges, and *PROC MIXED* often fail to converge, the two sets of results were compared and were found to be virtually identical when both *PROC IML* and *PROC MIXED* converge.

Example 1. (Dental Data A): This data set given in Table 1, taken from Timm (1980, Table 7.2) has nine subjects. The data were originally collected by T. Zullo of the School of Dental Medicine at the University of Pittsburgh. Three measurements at three different time points ($p = 3$) were made on each of ($q = 3$) characteristics. Note that the null Hypothesis cannot be tested as sample size is small and the number of subjects $n = 9$ is not greater than $pq = 9$. Thus, for the purpose of illustration, we will consider only two variables. For the first two variables in the Dental Data A, By our proposed algorithm we get $-2\ln\Lambda = 67.5486$ with 17 degrees of freedom; see Appendix 1. The corresponding $p-$value is negligible.

Table 1. Dental Data A (Timm, 1980, Table 7.2)

	Measurement 1			Measurement 2			Measurement 3		
Sub	Time 1	Time 2	Time 3	Time 1	Time 2	Time 3	Time 1	Time 2	Time 3
1	117	117.5	118.5	59	59	60	10.5	16.5	16.5
2	109	110.5	111	60	61.5	61.5	30.5	30.5	30.5
3	117	120	120.5	60	61.5	62	23.5	23.5	23.5
4	112	126	127	67.5	70.5	71.5	33	32	32.5
5	116	118.5	119.5	61.5	62.5	63.5	24.5	24.5	24.5
6	123	126	127	65.5	61.5	67.5	22	22	22
7	130.5	132	134.5	68.5	69.5	71	33	32.5	32
8	126.5	128.5	130.5	69	71	73	20	20	20
9	113	116.5	118	58	59	60.5	25	25	24.5

We may therefore conclude that the null hypothesis is statistically significant. Using *PROC MIXED* we get $-2\ln\Lambda = 262.4 - 194.9 = 67.5$. where 262.4 and 194.9 are the values of "-2 Log Likelihood" reported by SAS for the two models under H_o and H_a respectively. See the output as reported in Appendix 3. The above test statistic under H_o, follows a chi-square distribution with degrees of freedom ν, where ν is computed as $\nu = 20 - 3 = 17$, the same conclusion as we have drawn by our proposed method. For the first and third variables and also for second and third variables both our algorithm and *PROC MIXED*

give the same value of the test statistics $-2\ln\Lambda$ and the corresponding degrees of freedoms. See Table 3.

Example 2. (Dental Data B): The data set given in Table 2, is taken from Timm (1980, Table 7.2). Three measurements at three different time points ($p = 3$) were made on each of ($q = 3$) characteristics. Data were first analyzed only for the first two variables. The calculated value of the test statistic $-2\ln\Lambda = 22.8021$, with degrees of freedom 17. The corresponding $p-$value is 0.1558. So, we may accept the null hypothesis in this case. The *PROC MIXED* did not converge due to its problem of lack of convergence (LC) of the maximum likelihood estimates. For all other pairs of variables, say for the first and third variables and also for second and third variables, our proposed algorithm calculates the value of the test statistic $-2\ln\Lambda$ (Table 3) and the corresponding degrees of freedom. As earlier, *PROC MIXED* did not converge in any of these cases.

Table 2. Dental Data B (Timm, 1980, Table 7.2)

	Measurement 1			Measurement 2			Measurement 3		
Pat #	Time 1	Time 2	Time 3	Time 1	Time 2	Time 3	Time 1	Time 2	Time 3
1	128	129	131.5	67	67.5	69	24	24	24
2	116.5	120	121.5	63.5	65	66	28.5	29.5	29.5
3	121.5	125.5	127	64.5	67.5	69	26.5	27	27
4	109.5	112	114	54	55.5	57	18	18.5	19
5	133	136	137.5	72	73.5	75.5	34.5	34.5	34.5
6	120	124.5	126	62.5	65	66	26	26	26
7	129.5	133.5	134.5	65	68	69	18.5	18.5	18.5
8	122	124	125.5	64.5	65.5	66	18.5	18.5	18.5
9	125	127	128	65.5	66.5	67	21.5	21.5	21.6

Example 3. (Osteoporosis Data): The third data set corresponds to 19 subjects with osteoporosis. The measurements on bone mineral density are taken on $q = 4$ different body positions (characteristics), namely, spine, radius, femoral neck and hip, each at $p = 5$ time points at an interval of six months. The

Table 3 Comparison of $-2\ln\Lambda$ Between the Proposed Algorithm and PROC MIXED

Data	Variables Taken	Our Algorithm	PROC MIXED
Dental A	1,2	67.5486	67.5
Dental A	1,3	77.6394	77.7
Dental A	2,3	64.1587	64.2
Dental B	1,2	22.8021	LC ‡
Dental B	1,3	38.4535	LC
Dental B	2,3	45.0659	LC
Osteoporosis	1,2,3	274.1432	274.2
Osteoporosis	1,3,4	273.6982	273.7
Osteoporosis	1,2,4	288.3039	288.3
Osteoporosis	2,3,4	309.4814	LC

‡ LC= Lack of Convergence.

actual data are not provided here. We first consider the data set with only first three variables, namely, the measurements of bone mineral density at spine, radius and femoral neck. Thus $p = 5$ and $q = 3$, so that $pq = 15$. Clearly $n = 19$. The test statistic, $-2\ln\Lambda$ is equal to 274.1432 with 113 d.f. The corresponding $p-$value is negligible. Therefore, we may reject the null hypothesis. Using *PROC MIXED* we get the test statistic value as 274.2 with d.f. 113.

When we consider other triplets of variables except 2^{nd}, 3^{rd} and the 4^{th} variables, the test statistic values, $-2\ln\Lambda$ by our proposed algorithm using *PROC IML* and by using *PROC MIXED* are same

(Table 3). But in case 2nd, 3rd and the 4th variables, to test the null hypothesis, the test statistic value $-2\ln\Lambda$ by our proposed method is equal to 309.4814, but this time *PROC MIXED* failed to converge. We notice that whenever, *PROC MIXED* converges, the test statistic value $-2\ln\Lambda$ results the same value of the test statistic by our proposed algorithm, except for a minor round off.

CONCLUSIONS

Objective of this note was to provide a usable SAS algorithm to test the Kronecker product covariance structure hypothesis using the likelihood ratio test. The algorithm presented here successfully completes the task for all the data sets that we tried. While we have confined our alternative to the case when the repeated measures have AR(1) structure, the program can be suitably modified for other structures as well. More details about certain mathematical expression needed for such modification can be found in Roy (2002).

APPENDIX

APPENDIX 1 (SAS IML CODES FOR OUR PROPOSED ALGORITHM)

```
/* The program calculates the test statistic -2 Log Likelihood (chi_sq)
and the corresponding p-val of the proposed test. The program calculates
the estimate S of the unstructured variance covariance matrix Omega in Zullo's
the Dental Data set A (Timm (1980, Table 7.2)). We have taken the first two
variables i.e. q=2, three repetitions of each variable i.e. p=3. */

options nocenter ls=80 ps=50 nodate nonumber;
data a;
infile 'c:\sugi_paper\zuloA.dat';
input sub x1-x9;

proc iml symbolsize=3036;
converge=0.000001;  pi=3.14159265;
c1={0 0 0,
    0 1 0,
    0 0 0};
c2={0 1 0,
    1 0 1,
    0 1 0};
use a;
read all var {x1 x2 x4 x5 x7 x8} into gp;
read all var {x1 x4 x7} into gp1;
read all var {x2 x5 x8} into gp2;

n=nrow(gp); q=2; p=3;
one={1,1,1}; jj1={1,0,0}; jj2={0,1,0}; jj3={0,0,1};
j0=j(q,q,0); j1=I(q)||j0||j0; j2=j0||I(q)||j0; j3=j0||j0||I(q);

/* gpbar is  the means of the Dental Data A. */
gpbar=gp[+, ]/n; gp1bar=gp1[+, ]/n; gp2bar=gp2[+, ]/n;

w=j(p*q,p*q,0);
do j=1 to n;
   use a; read all var {x1 x2 x4 x5 x7 x8} into subb where (sub=j);
   w=w+t(subb-gpbar)*(subb-gpbar);
end;
detw=det(w);
```

```
gp1barm=repeat(gp1bar,n,1);
gp1a=gp1-gp1barm; gp1ass=gp1a[##, ];
sq_gp1assm= repeat(sqrt(gp1ass),n,1);
gp1an=gp1a/ sq_gp1assm;
vi1=t(gp1an)*gp1an;

gp2barm=repeat(gp2bar,n,1);
gp2a=gp2-gp2barm; gp2ass=gp2a[##, ];
sq_gp2assm= repeat(sqrt(gp2ass),n,1);
gp2an=gp2a/ sq_gp2assm;
vi2=t(gp2an)*gp2an;
vi=(vi1+vi2)/2;

/* Calculating the initial estimate Ve of V. rest is the common estimate
of rho in both the classes  */
g1=(t(jj1)*vi*jj2+t(jj2)*vi*jj3)/(p-1);
g2=(t(jj1)*vi*jj3)/(p-2);
po=p-1; go=p-1;
If g1<0 then do;
   go=go-1;
   g1=0;
end;
If g2<0 then do;
   go=go-1;
   g2=0;
end;
rest=((g1**po+g2**(po/2))/go)**(1/po);
d=rest; d1=d; d2=d1*d;
d11=1||d1||d2;
d12=d1||1||d1;
d13=d2||d1||1;
ve=d11//d12//d13;

/* Calculating the mle's  mlv and mlsig of V and sigma respectivaly in
population 2.*/
mlsige=j(q,q,0); iter=0;
do until (maxab<converge);
   ive=inv(ve);
   v11=t(jj1)*ive*jj1; v12=t(jj1)*ive*jj2; v13=t(jj1)*ive*jj3;
   v21=t(jj2)*ive*jj1; v22=t(jj2)*ive*jj2; v23=t(jj2)*ive*jj3;
   v31=t(jj3)*ive*jj1; v32=t(jj3)*ive*jj2; v33=t(jj3)*ive*jj3;

   sigma2e=j(q,q,0);
   do j=1 to n;
      use a;
      read all var {x1 x2 x4 x5 x7 x8} into subb where (sub=j);
      subba=subb-gpbar;
      sub1=j1*t(subba); sub2=j2*t(subba); sub3=j3*t(subba);
      sigma2e=sigma2e+v11*((sub1)*t(sub1))+v21*((sub1)*t(sub2))+
              v31*((sub1)*t(sub3))+
              v12*((sub2)*t(sub1))+v22*((sub2)*t(sub2))+
              v32*((sub2)*t(sub3))+
              v13*((sub3)*t(sub1))+v23*((sub3)*t(sub2))+
              v33*((sub3)*t(sub3));
   end;
   mlsig=sigma2e/(n*p);
   absig=abs(trace(mlsige-mlsig));
```

```
   mlsige=mlsig;
   imlsig=inv(mlsig);
   kk3=trace((I(p)@imlsig)*w);  mk3=trace((c1@imlsig)*w);  nk3=trace((c2@imlsig)*w);

   /* solving the cubic equation */
   s=p-1;
   pp=(nk3)/(-2*n*s*q); qq=(2*n*s*q-2*kk3-2*mk3)/(-2*n*s*q); rr=nk3/(-2*n*s*q);
   aa=(1/3)*(3*qq-pp**2); bb=(1/27)*(2*pp**3 -9*pp*qq +27*rr);
   discrim=(bb**2)/4+(aa**3)/27;
   if discrim>0 then
      do;
         s1=((bb**2)/4+(aa**3)/27)**0.5;
         s2=(-bb/2+ s1)**(1/3);
         ar=-bb/2-s1;
         if ar <0 then
            do;
               ar1=-ar; ar2=ar1**(1/3); s3=-ar2;
            end;
         else
            do;
               s3=ar**(1/3);
            end;
         ro2=s2+s3-pp/3;
         d=ro2; d1=d; d2=d1*d;
         d11=1||d1||d2;
         d12=d1||1||d1;
         d13=d2||d1||1;
         mlv=d11//d12//d13;
         ve=mlv;
      end;
   if discrim <0 then
      do;
         dm=sqrt(-discrim);
         r=sqrt( (bb**2)/4+dm**2);
         th=atan(-2*dm/bb);  /*print r th;*/
         pplq=2*(r**(1/3))*cos(th/3); pmiq=2*(r**(1/3))*sin(th/3);
         rt1=pplq-pp/3; rt2=-0.5*pplq-pp/3-0.5*pmiq*sqrt(3);
         rt3=-0.5*pplq-pp/3+0.5*pmiq*sqrt(3);
         if rt1>1 then rt1=0;
         if rt2>1 then rt2=0;
         if rt3>1 then rt3=0;
         rtvec=rt1//rt2//rt3;
         ro2=max(rtvec);
         d=ro2; d1=d; d2=d1*d;
         d11=1||d1||d2;
         d12=d1||1||d1;
         d13=d2||d1||1;
         mlv=d11//d12//d13;
         ve=mlv;
      end;
   iter=iter+1;
   abr1=abs(rest-ro2);
   rest=ro2;
   maxab=max(absig//abr1);
end;
ew1=0.5*trace(inv(ve@mlsig)*w);
detsig=(det(mlsig));
```

```
detve=det(ve);
lrnu=detve**(-q*n/2) * detsig**(-p*n/2);
lrde=detw**(-n/2) *n**(n*p*q/2); de=0.5*n*p*q;
lr=(lrnu/lrde)* exp(de-ew1);
chi_sq=-2*(log(lrnu/lrde)+ (de-ew1) );
df=(p*q*(p*q+1)/2)-(q*(q+1)/2)-1;
p_val=1-probchi(chi_sq, df);
print df chi_sq p_val;
run;
```

APPENDIX 2 (SAS PROC MIXED CODES)

```
/* The program fits the linear mixed effects model under the null
hypothesis Omega= UN @ AR(1) for the first two variables of the Dental
data.  As a byproduct we get -2 Log Likelihood in the output.*/

options nocenter ls=80 ps=50 nodate nonumber;
data a;
infile 'c:\sugi_paper\zuloA12.dat';
input x1-x6;
data b; set a;
subj=_n_;
x=x1; m_var='var1'; time=1; output;
x=x2; m_var='var2'; time=1; output;
x=x3; m_var='var1'; time=2; output;
x=x4; m_var='var2'; time=2; output;
x=x5; m_var='var1'; time=3; output;
x=x6; m_var='var2'; time=3; output;
drop x1-x6;
proc mixed data=b method=ml covtest;
classes subj m_var time;
model x=m_var time m_var*time;
repeated m_var time/type= un@ar(1) subject=subj;
Title '-2 Log Likelihood under the Null Hypothesis Ho';


/* This part of the program fits the model under the alternative
hypothesis Omega = UN for the first two variables of the Dental
data. As a byproduct here also we get -2 Log Likelihood in the  output. */

data c; set a;
array t{6} x1-x6;
subj+1;
do tt=1 to 6;
   x=t{tt};
   output;
end;
drop x1-x6;
run;
proc mixed data=c method=ml covtest;
classes subj tt;
model x=tt;
repeated /type= un subject=subj;
Title '-2 Log Likelihood under the Alternative Hypothesis Ha';
run;
```

APPENDIX 3 (OUTPUT OF PROC MIXED)

```
-2 Log Likelihood under the Null Hypothesis Ho

          Fit Statistics

-2 Log Likelihood               262.4
AIC  (smaller is better)        282.4
AICC (smaller is better)        287.5
BIC  (smaller is better)        284.4

  Null Model Likelihood Ratio Test

    DF    Chi-Square      Pr > ChiSq

     3      79.02            <.0001
```

```
-2 Log Likelihood under the Alternative Hypothesis Ha

          Fit Statistics

-2 Log Likelihood               194.9
AIC  (smaller is better)        248.9
AICC (smaller is better)        307.0
BIC  (smaller is better)        254.2

  Null Model Likelihood Ratio Test

    DF    Chi-Square      Pr > ChiSq

    20      146.57           <.0001
```

REFERENCES

Boik, J. B. (1991), "Scheffe's Mixed Model for Multivariate Repeated Measures: A Relative Efficiency Evaluation," *Commun. Statist.– Theory Meth.*, 20, 1233-1255.

Chaganty, N. R. and Naik, D. N. (2002), "Analysis of Multivariate Longitudinal Data using Quasi-Least Squares," *Journal of Statistical Planning and Inference*, 103, 421-436.

Galecki, A. T. (1994), "General Class of Covariance Structures for Two or More Repeated Factors in Longitudinal Data Analysis," *Commun. Statist.– Theory Meth.*, 23, 3105-3120.

Khattree, R. and Naik, D.N. (1999), *Applied Multivariate Statistics with SAS Software*, 2nd Ed., Wiley, New York, NY.

Naik, D. N. and Rao, S. (2001), "Analysis of Multivariate Repeated Measures Data with a Kronecker Product Structured Covariance Matrix," *J. Appl. Statist.*, 28, 91-105.

Roy, A. (2002), *Some Contributions to Discrimination and Classification with Repeated Measures Data with Special Emphasis on Biomedical Applications*, Ph.D. Dissertation, Department of Mathematics and Statistics, Oakland University, Rochester, MI.

Roy, A. and Khattree, R. (2004), "On Implementation of a Test for Kronecker Product Covariance Structure for Multivariate Repeated Measures Data," Submitted.

Roy, A. and Khattree, R. (2005), "On Discrimination and Classification with Multivariate repeated Measures Data," *Journal of Statistical Planning and Inference*, In Press.

Roy, A. and Khattree, R. (2003), "Tests for Mean and Covariance Structures Relevant in Repeated Measures Based Discriminant Analysis," *Journal of Applied Statistical Science*, 12(2), 91-104.

SAS Institute Inc. (2004), *SAS/STAT User's Guide Version 9,* Cary, NC: SAS Institute Inc.

Timm, N. H. (1980), "Multivariate Analysis of Variance of Repeated Measurements," In P.R.Krishnaiah (Ed.), *Handbook of Statistics, Vol. 1* North-Holland, 41-87.

CONTACT INFORMATION

The authors welcome comments and suggestions. Please direct enquiries to:
Anuradha Roy
Department of Management Science and Statistics
The University of Texas at San Antonio
6900 N Loop 1604 West, San Antonio, TX 78249
Tel: (210)458-6343
E-mail: aroy@utsa.edu